Pedestrian lights that operate temporarily when necessary 

Main Idea
The idea to implement a simple controlling circuit, using a TI Innovator in a vivid and close to real life example. Pedestrian lights that operate temporarily when necessary should be installed at a pedestrian crossing. The traffic lights should switch to the red light and after actuating a switch, subsequently to the green light again.
[image: IMG_2236]
Overview
· The traffic lights are simulated with simple LED lamps.
· The trigger signal for changing the light is tripped by a switch
· The program is required to work continuously, a brightness sensor is used to stop the process. 

[image: ]Hardware
In addition to a TI Handheld, the TI Innovator and a patch panel, a switch for patch panels, 5 LEDs (2 red ones, a yellow one and 2 green ones) as well as some jumper cables with male connectors are needed.


Structural  Design
[image: IMG_2237a]
On the patch panel, the LEDs are plugged in in two groups in order to simulate both traffic lights. 
The cathodes of the LEDs can be plugged into the same line of the patch panel and are afterwards connected to a ground jack of the TI Innovator (bottom line, pins 1-8) with a cable. 
The anodes of the LEDs are connected with the jack socket  of the top line each. The red LED of the traffic light is being connected to PIN BB1, the yellow LED to BB2 and the green LED to BB3. 
Next, the switch is connected to BB7 and ground and put in a way on the patch panel that the connecting line of the plastic fittings on the the bottom side is perpendicular to the lines of the panel. 
[image: Ampel BB]


Structure of the Program 

Basically, a continuous run of the program should be possible. 
In order to prevent an infinite loop, a break condition has to be defined. The brightness sensor, which is integrated in the TI Nspire, could for instance be used for that purpose. 
In a first attempt, the brightness of the surroundings is measured. A program run is carried out in normal light conditions, the program is stopped at darkness. The program can therefore be run continuously and stopped by simply covering the sensor with a fingertip. 

Query concerning the country the traffic light is to be used
There is a difference concerning the end of the green phase of traffic lights in Austria and Germany. In contrast to Austria, where the green light starts flashing towards the end of the green phase, the latter is ended by a sudden change to the yellow light in Germany. The program can simulate both options and the corresponding mode can be chosen any time the program starts. 

Source code
Due to the operating of each LED and redundancies in the code, which support the clarity as well as the stability of the program, the code reaches a certain length. In this example, it was tried to use several kinds of loops in order to show differences. The complete source code can be found on the following page.

Additional Tasks
The project offers some possibilities to carry out both simpler and more demanding tasks. As a simplification a autonomous traffic light without interaction by pedestrians could be done, controlled by time intervals for the different conditions.
[bookmark: _GoBack]Instead of a simple breadboard a real intersection with four traffic lights and the according pedestrian lights could be built and controlled by the TI innovator.

Define ampel()=
Prgm
:Send "CONNECT BUTTON 1 TO BB7"
:Send "CONNECT LED 1 TO BB1"
:Send "CONNECT LED 2 TO BB2"
:Send "CONNECT LED 3 TO BB3"
:Send "CONNECT LED 4 TO BB4"
:Send "CONNECT LED 5 TO BB5"
:n:=5
:b:=0
:mode:=0
:c:=0
:© Abfrage Land
:Disp "Bitte wählen Sie den Anzeigemodus"
:Disp "1 für Deutschland"
:Disp "2 für Österreich"
:Request "Modus 1=BRD / 2=AT: ",mode
:  Loop
:    If mode=1 or mode=2 Then
:      Exit
:    Else
:      Disp "Eingabe ungültig!"
:      Request "Modus 1=BRD / 2=AT: ",mode
:    EndIf
:  EndLoop
:
:© Modus für Österreich:   Grünphase endet mit Blinken
:If mode=2 Then
:  Disp "Ampelmodus für Österreich gewählt"
:  Disp "Bitte roten Taster drücken!"
:  Disp ""
:  Loop
:© Der Helligkeitssensor wird als "Schalter" für den Programmabruch genutzt
:    If n>0.6 Then
:      Send "READ BRIGHTNESS"
:      Get n
:      Wait 0.1
:      Send "READ BUTTON 1"
:      Get b
:      Wait 0.1
:      If b=0 Then
:© Fußgängerampel zeigt rot
:        Send "SET LED 1 OFF"
:        Send "SET LED 2 OFF"
:        Send "SET LED 3 ON"
:        Send "SET LED 4 ON"
:        Send "SET LED 5 OFF"
:        Wait 0.5
:      Else
:© Schalter betätigt, Umschaltsequenz wird abgearbeitet  
:        Disp "Signal kommt, bitte warten..."
:        b:=0
:        c:=c+1
:        Disp "Fußgängerphase Nummer ",c," startet!"
:        Disp ""
:        Send "SET LED 1 OFF"
:        Send "SET LED 2 OFF"
:        Send "SET LED 3 ON"
:        Send "SET LED 4 ON"
:        Send "SET LED 5 OFF"
:        Wait 0.5
:© Grünblinken-Straße
:        m:=0
:        While m<4
:          m:=m+1
:          Send "SET LED 3 OFF"
:          Wait 0.3
:          Send "SET LED 3 ON"
:          Wait 0.5
:        EndWhile
:        Send "SET LED 3 OFF"
:        Send "SET LED 2 ON"
:        Wait 1
:        Send "SET LED 1 ON"
:        Send "SET LED 2 OFF"
:        Send "SET LED 3 OFF"
:        Wait 0.5
:        Send "SET LED 4 OFF"
:        Send "SET LED 5 ON"
:        Wait 2
:© Grünblinken-Fußgänger
:        For i,1,4
:          Send "SET LED 5 OFF"
:          Wait 0.3
:          Send "SET LED 5 ON"
:          Wait 0.5
:        EndFor
:        Send "SET LED 4 ON"
:        Send "SET LED 5 OFF"
:        Wait 0.5
:        Send "SET LED 2 ON"
:        Wait 1
:        Send "SET LED 1 OFF"
:        Send "SET LED 2 OFF"
:        Send "SET LED 3 ON"
:        Send "SET LED 4 ON"
:        Send "SET LED 5 OFF"
:        Wait 0.5
:        Disp "Bitte roten Taster drücken!"
:      EndIf
:    Else
:        Exit
:    EndIf
:  EndLoop
:Else
:  Disp "Ampelmodus für Deutschland gewählt"
:  Disp "Bitte roten Taster drücken!"
:  Disp ""
:  Loop
:    If n>0.6 Then
:      Send "READ BRIGHTNESS"
:      Get n
:      Wait 0.1
:      Send "READ BUTTON 1"
:      Get b
:      Wait 0.1
:      If b=0 Then
:        Send "SET LED 1 OFF"
:        Send "SET LED 2 OFF"
:        Send "SET LED 3 ON"
:        Send "SET LED 4 ON"
:        Send "SET LED 5 OFF"
:        Wait 0.5
:      Else
:        Disp "Signal kommt, bitte warten..."
:        b:=0
:        c:=c+1
:        Disp "Fußgängerphase Nummer ",c," startet!"
:        Disp ""
:        Send "SET LED 1 OFF"
:        Send "SET LED 2 OFF"
:        Send "SET LED 3 ON"
:        Send "SET LED 4 ON"
:        Send "SET LED 5 OFF"
:        Wait 0.5
:        Send "SET LED 3 OFF"
:        Send "SET LED 2 ON"
:        Wait 1
:        Send "SET LED 1 ON"
:        Send "SET LED 2 OFF"
:        Wait 0.5
:        Send "SET LED 4 OFF"
:        Send "SET LED 5 ON"
:        Wait 2
:        For m,1,4
:          Send "SET LED 5 OFF"
:          Wait 0.3
:          Send "SET LED 5 ON"
:          Wait 0.5
:        EndFor
:        Send "SET LED 4 ON"
:        Send "SET LED 5 OFF"
:        Wait 0.5
:        Send "SET LED 2 ON"
:        Wait 1
:        Send "SET LED 1 OFF"
:        Send "SET LED 2 OFF"
:        Send "SET LED 3 ON"
:        Send "SET LED 4 ON"
:        Send "SET LED 5 OFF"
:        Wait 0.5
:        Disp "Bitte roten Taster drücken!"
:      EndIf
:    Else
:      Exit
:    EndIf
:  EndLoop
:EndIf
:Send "DISCONNECT BUTTON 1 TO BB8"
:Send "DISCONNECT LED 1"
:Send "DISCONNECT LED 2"
:Send "DISCONNECT LED 3"
:Send "DISCONNECT LED 4"
:Send "DISCONNECT LED 5"
:EndPrgm

